Supplemental File 2

SAS code for mixed model mapping of IBM QTL in single and combined-stress environments, written by G. Buddhika Makumburage, with assistance from Ann E. Stapleton and James B. Holland.
Step 1
/*specifies Path to the data sets;*/

libname lsIBM94 'C:\LSIBM\raw data'; 

data envi;/*imports response variables data set*/


set lsIBM94.sheet1;

run;

proc sort data=lsIBM94.markerset out=markerset;/*sorts independent variables by marker*/


by marker;

run;

/*makes appropriate transpose to independent variable data set */

proc transpose data=markerset(keep=marker MO001 - - MO384) out=markertrans;


by marker;


var MO001--MO384;/*these form the range of columns for the lines*/

run;

proc sql;/*joins dependent and independent variables */


create table joined as


select Markertrans.marker,Markertrans._NAME_ ,Markertrans.COL1 as marker_state,


Envi.Levene_meanRatio,Envi.Env


from Markertrans,Envi


where (Markertrans._NAME_ eq Envi.Line)


;

quit;

proc sort data=joined;/*sorts the 'joined' data set by marker*/


by marker;

run; 

/*Run these piece of codes( The 5 lines below ) only if you want to write sas log to a file*/

options nonotes nosource nosource2 errors=0;

proc printto log="C:\LSIBM\SAS Log\filename.log";

run;

proc printto print="C:\LSIBM\SAS Log\filename.lst";

run;

/*runs proc mixed for the joined data set*/

proc mixed data=joined covtest method=type3 ;


by marker;


where marker_state in ('A','B');


class Env marker_state  _NAME_ ;


model Levene_meanRatio=    marker_state  /s ;/*specifies marker state to be fixed effect*/


random _NAME_(marker_state) Env Env*marker_state /s CL;/*specifies random effects.  _NAME_ represents the line of the plant */



lsmeans marker_state;


ods output tests3=tests lsmeans=means SolutionR=REfect SolutionF=FEfect CovParms=covtest 

                type3=randomType3;

run;

quit;

/*run these 3 lines below only if you wrote sas log to a different file*/

proc printto;

run;

options notes source source2 errors=20;

/*specifies path to save important sas data sets created by proc mixed and proc sql*/

libname results 'C:\LSIBM\outputStep1';

data results.tests;/* saves test data set on the location*/

set tests;

run;

data results.means;/* saves means data set on the location*/

set means;

run;

data results.REfect;/* saves Refect data set on the location*/

set REfect;

run;

data results.FEfect;/* saves Fefect data set on the location*/

set FEfect;

run;

data results.covtest;/* saves covtest data set on the location*/

set covtest;

run;

data results.Randomtype3;/* saves Randomtype3 data set on the location*/

set Randomtype3;

run;

Step 2

/*specifies location for the data sets produced by step 1 */

libname CHeigth 'C:\LSIBM\outputStep1';

data randomType3;/*imports randomType3 data set created from step1*/

set CHeigth.randomType3;

if  Source ne 'Env*marker_state'  then delete;

run;

/*location for raw data sets */

libname IBM94 'C:\LSIBM\raw data';

proc sort data=IBM94.markerset out=markerset;/*sorts markerset by location */

by increment;/*increment represents the location of a marker */

run;

data markerset;/*creates location index for markers*/

   set markerset;

   dummy=_n_;/*dummy represents location index*/

   run;

   proc sql;/*adds location to randomtype3 data set*/

   create table pvalcorlnorder as

   select randomType3.*, markerset.dummy 

   from randomType3, markerset

   where markerset.marker= randomType3.marker;

   run;

   quit;

/*creates fractional rank for the pvalues*/

proc rank data=pvalcorlnorder out=pvalcorlnorder2 fraction;


var probF;


ranks rankp;

run;

  proc sort data= pvalcorlnorder2;/*sorts fractional ranks by location index*/

  by dummy;

  run;

/*adjusts fractional ranks for multitest using proc psmooth*/

proc psmooth data=pvalcorlnorder2 out=pnew simes bandwidth=3 to 9 by 2 neglog;

      var rankp;

      id dummy;/*orders values by location index*/

   run;

   symbol1 v=none i=join;

   legend1 label=none;

    ods listing;

   proc gplot data=pnew;/*creates plots from psmooth output*/

      plot (rankp rankp_S3 rankp_S5 rankp_S7 rankp_S9)*dummy

           /overlay vref=3.0 legend=legend1;

   run;

data pnewedit;/*keeps significat markers at 5% level for further analysis*/

set pnew;

if rankp_S3 <= 2.995732

 then delete;

run;

data refect1;/*imports refect data set created from step1*/

set CHeigth.refect;

if Effect ne 'Env*marker_state' then delete;/*keeps estimates only from interaction*/

pot= cats(Env,marker_state);

keep marker Effect Lower Estimate upper pot;

run;

/*makes appropriate transpose to refect1 data to be more readable for grouping purpose*/

proc transpose data=refect1(keep=marker Effect Lower Estimate upper pot) out=refect2;


by marker;


var Estimate Lower  upper;

run;

data refectEs;/*detects only estimates for levels of interaction from refect2*/

set refect2;

if _NAME_ ne 'Estimate' then delete;

rename COL1= E1A  COL2= E1B COL3= E2A  COL4= E2B COL5= E3A COL6= E3B COL7= E4A COL8= E4B;

run;

data refectLw;/*detects only lower confidence intervals for levels of interaction 

                  from refect2*/

set refect2;

if _NAME_ ne 'Lower' then delete;

rename COL1= L1A  COL2= L1B COL3= L2A  COL4= L2B COL5= L3A COL6= L3B COL7= L4A COL8=L4B;

run;

data refectUp;/*detects only upper confidence intervals for levels of interaction

               from refect2*/

set refect2;

if _NAME_ ne 'Upper' then delete;

rename COL1= U1A  COL2= U1B COL3= U2A  COL4= U2B COL5= U3A COL6=U3B COL7= U4A COL8=U4B;

run;

data refect;/*combines estimates, lower & upper CI for levels of interaction*/

retain marker E1A L1A U1A E1B L1B U1B E2A L2A U2A E2B L2B U2B E3A L3A U3A E3B L3B U3B E4A L4A U4A E4B L4B U4B;

merge refectEs refectLw refectUp;

by marker;

drop _NAME_;

run;

/*combines any other important details for the significant markers for grouping*/

proc sql;

create table signfcnttmarker as

select pnewedit.dummy,refect.* , markerset.marker,markerset.increment, 

markerset.Chromosome, pvalcorlnorder2.FValue,pvalcorlnorder2.ProbF,

pvalcorlnorder2.rankp 

from pnewedit,pvalcorlnorder2,markerset,refect

where (pnewedit.dummy = pvalcorlnorder2.dummy)& (pnewedit.dummy = markerset.dummy)

      & (refect.marker= markerset.marker) ;

run;

quit;

/*creates groupmarker data set from signfcnttmarker*/

data groupmarker;

set signfcnttmarker;

run;

/*groups markers manually by considring location and confidence intervals look at groupmarker table and type the dummy values into the set of if statements below */

data groupmarker;

set groupmarker;

if 716<=dummy<=721 then group=1; /*dummy represents location of marker*/

if 891<=dummy<=896 then group=2;

if 1874<=dummy<=1880 then group=3;

if 1907<=dummy<=1917 then group=4;

if 1977<=dummy<=2011 then group=5;

if 2240<=dummy<=2250 then group=6;

if 2301<=dummy<=2303 then group=7;

if 2320<=dummy<=2324 then group=8;

if 2505<=dummy<=2523 then group=9;

if 2695<=dummy<=2701 then group=10;

if 2938<=dummy<=2942 then group=11;

run;

/*saves a csv file of group details*/

proc export data=groupmarker outfile='C:\LSIBM\outputStep2\csvFiles\GroupDetails.csv'

replace;

run;

/*imports response variables data set*/

data envi;


set IBM94.sheet1;

run;

proc sort data=markerset;/*sorts independent variables by marker*/


by marker;

run;

/*makes appropriate transpose to independent variable data set */

proc transpose data=markerset(keep=marker MO001 - - MO384) out=markertrans;


by marker;


var MO001--MO384;/*these form the range of columns for the lines*/

run;

proc sql;/*joins dependent and independent variables */


create table joined as


select Markertrans.marker,Markertrans._NAME_,Markertrans.COL1 as marker_state,


Envi.Levene_meanRatio,Envi.Env


from Markertrans,Envi


where (Markertrans._NAME_ eq Envi.Line)


;

quit;

proc sql;/*subsets joined data set with the group markers*/

create table subjoined as

select joined.*,groupmarker.group

from joined,groupmarker

where joined.marker=groupmarker.marker;

run;

quit;

proc sort data=subjoined;/*sorts the data by group*/

by group;

run;

proc mixed data=subjoined  covtest method=type3 ;/*runs proc mixed by group */


by group;


where marker_state in ('A','B');


class Env marker_state  _NAME_ ;


/*specifies marker state to be fixed effect*/


model Levene_meanRatio= marker_state  /s OUTPRED=predict;


/*specifies random effects._NAME_ represents the line of the plant */


random _NAME_(marker_state) Env Env*marker_state /s CL ;



lsmeans marker_state;

   /*performs comparisons between levels of interaction*/


estimate'2A vs 4A' | Env*marker_state  1 0 0 0 -1 0;


estimate'2B vs 4B' | Env*marker_state 0 1 0 0 0 -1;


estimate'3A vs 4A' | Env*marker_state 0 0 1 0 -1 0;


estimate'3B vs 4B' | Env*marker_state  0 0 0 1 0 -1;


contrast '2A vs 4A' | Env*marker_state  1 0 0 0 -1 0;


contrast'2B vs 4B' | Env*marker_state 0 1 0 0 0 -1;


contrast'3A vs 4A' | Env*marker_state 0 0 1 0 -1 0;


contrast'3B vs 4B' | Env*marker_state  0 0 0 1 0 -1;


ods output tests3=tests lsmeans=means SolutionR=REfect SolutionF=FEfect 

      CovParms=covtest type3=randomType3 estimates=estmte Contrasts=Contrst;

run;

quit;

/*specifies path to save important sas data sets*/

libname group 'C:\LSIBM\outputStep2\sasdataset';

data group.tests;/* saves test data set on the location*/

set tests;

run;

data group.means;/* saves means data set on the location*/

set means;

run;

data group.REfect;/* saves Refect data set on the location*/

set REfect;

run;

data group.FEfect;/* saves Fefect data set on the location*/

set FEfect;

run;

data group.covtest;/* saves covtest data set on the location*/

set covtest;

run;

data group.Randomtype3;/* saves Randomtype3 data set on the location*/

set Randomtype3;

run;

data group.compareEstmte;/* saves estmte data set on the location*/

set estmte;

run;

data group.Contarst;/* saves Contrast data set on the location*/

set Contrst;

run;

data forplot;/*detects only interaction estimates for plots*/

set refect;

if Effect ne 'Env*marker_state' then delete;

xval= cats(Env,marker_state);/*identifies the level of interaction*/

run;

/*saves a csv file of forplot data*/

proc export data=forplot outfile='C:\LSIBM\outputStep2\csvFiles\interaction.csv' replace;

run;

/*creates interaction plots for groups*/

 goptions reset=all  colors=(black blue green red)  ftitle=swissb ;  

 symbol1 color=red  value=dot height=.2 ;  

ods listing;

ods rtf file = "interactionplot(ChangeInHeight).rtf";/*creates word document of plots*/

 %macro plot;/*generates plots */

%do j=1 %to 12;/*creates plots for all groups at once through the loop*/

data _null_;

set Groupmarker(where=(group=&j));

by group;

 if first.group then  call symput('first',put(dummy,5.));

 if last.group then call symput('last',put(dummy,5.));

run;

%if &first=&last %then %let lastname= &last;

%if &first ne &last %then %let lastname= &first to &last;

%put &lastname;

title "Interaction Estimates: group= &j , Location Index: &lastname ";

proc gchart data=forplot(where=(group= &j));

vbar xval  / TYPE=SUM sumvar=Estimate ; 

 run; 

quit;

 %end;

%mend;

%plot;

ods rtf close;/*finishes word document of plots*/

